

Program tapes from Nicolet Instruments Corporation are normally supplied in binary format. The 1080 must be programmed to read a binary tape. Initially, this is done by loading a simple program called the Short Binary Loader (SBL) through the 290 Switch Register (SR). Details of the operation of the 290 are explained elsewhere.

Once the SBL is in memory, a more sophisticated program labeled the "Binary Loader" may be loaded using the SBL. The Binary Loader performs all the functions of the SBL and, additionally, notifies the operator of a read-in error. Another feature of the Binary Loader is that it stops upon finding a "rubout" (377g) in the leader-trailer part of the tape.

These programs reside in core permanently and should not be destroyed. All NIC supplied programs avoid this area of core storage. If, however, they are accidentally modified, they can be restored by the following instructions.

It should be emphasized that both loaders read the same tape format. The SBL is used initially because it is shorter and therefore easier to load. Instructions for loading follow.

A. Entering the SBL Loader

1. Set the Switch Register (SR) to read 7753_g. The binary equivalent is 1 1 1 1 1 1 1 0 1 0 1 1.

NOTE: The switch position for a "1" is "up."

2. Depress the LOAD PC pushbutton.
3. Press EXECUTE. The two rows of lights labeled AC and PC should reflect the position of the SR.

4. The STEP button should be in the "in" position.
5. Place the DEPOSIT button in the "in" position.
6. Set the SR to the first number in the table on the following page.
7. Press EXECUTE. The row of lights labeled PC should increment and the row of lights labeled AC should reflect the SR.
8. Correspondingly, set each successive number from the table into the SR and press the EXECUTE button.
9. To verify: repeat steps 1 through 4, depress EXAMINE and then depress EXECUTE. Successive addresses are examined by continuing to depress the EXECUTE button and inspecting the AC display lights.
10. If an error is made, set the SR to the offended address and follow steps 2 through 7 paying attention that the correct corresponding instruction be used in step 6.

Once SBL is in memory, the Binary Loader can be loaded.

(Table follows on Page 3)

B. Entering the Binary Loader

This tape is entered using the SBL.

1. Set the Switch Register (SR) to 7770_8 (1 1 1 1 1 1 1 1 0 0 0).
See NOTE, Section C.
2. Depress the LOAD PC switch.
3. Depress the EXECUTE switch.
4. Place the Teletype switch in the LINE position.
5. Place the Tape Reader switch in the FREE position.

SBL LOADER

<u>ADDRESS</u>		<u>CONTENTS</u>							
<u>OCTAL</u>	<u>OCTAL</u>	<u>BINARY</u>							
7753	5007	00	000	000	101	000	000	111	
7754	6454	00	000	000	110	100	101	100	
7755	1754	00	000	000	001	111	101	100	
7756	4453	00	000	000	100	100	101	011	
7757	1001752	01	000	000	001	111	101	010	
7760	0	00	000	000	000	000	000	000	
7761	170000	00	001	111	000	000	000	000	
7762	2001752	10	000	000	001	111	101	010	
7763	2200	00	000	000	010	010	000	000	
7764	1770	00	000	000	001	111	111	000	
7765	2001752	10	000	000	001	111	101	010	
7766	2001752	10	000	000	001	111	101	010	
7767	1001760	01	000	000	001	111	110	000	
7770	2001760	10	000	000	001	111	110	000	
7771	2405776	10	100	000	101	111	111	110	
7772	2001760	10	000	000	001	111	110	000	
7773	3405776	11	100	000	101	111	111	110	
7774	2125776	10	001	010	101	111	111	110	
7775	1772	00	000	000	001	111	111	010	

6. Place the paper tape "BINARY LOADER NIC 80-7001-B" on the tape reader. The label should face up and extend from the front of the reader.
7. Move the tape until leader code (continuous holes on the right side of the tape) is above the reader head.
8. Place the Tape Reader switch in the START position.
9. Depress the CONTINUE switch on the 1080.
10. Press EXECUTE. At this point the tape should start moving and should continue through to the end of the tape. If it should stop at any point on the tape, check the SBL loader and start again at B.

C. Using the Binary Loader

1. Set the Switch Register (SR) to 7777_8 (1 1 1 1 1 1 1 1 1 1).
See NOTE this Section.
2. Depress the LOAD PC switch. The rows of lights labeled PC and AC should reflect the SR setting.
3. Place the Teletype switch to the LINE position.
4. Place the Tape Reader switch to the FREE position.
5. Load the desired Binary Program paper tape with the label up and extending from the front of the reader.
6. Move the Tape Reader switch to START.
7. Press the EXECUTE button on the 1080.
8. The tape should start moving and continue to move to the end of the tape. Three conditions can cause a halt. If the program fails to start and the Program Counter (PC) contains the number 7661_8 , the Binary Loader

program is in error. If this is the case, reload the Binary Loader using the SBL Loader (section B). If the tape stops at any point and rings a bell, the program detected an error. Check the tape for rips or folds and reload the tape. If the same symptom reoccurs, reload the SBL and RIM loaders. If the program reads a "rubout" while reading leader or trailer, the program will halt with 377_8 in the AC. This is a valid halt. If more tapes are to be read, go back to step 4 and repeat the following steps.

9. If no "rubout" or error was found, the tape read in correctly.
10. If more tapes are to be read, do steps 4 through 6.
11. If all the desired tapes are in memory, press the STOP button on the 1080.

The program just loaded should now be able to run. Starting instructions may differ; therefore, follow the directions provided with each program.

NOTE: For quick reference the starting address (SA) of a program is usually printed on the binary tape of that program. This is the starting address of the program when used and has no significance when loading the program.

For example the Binary Loader Program tape lists SA:7777 but the SBL when used to load the Binary Loader is started at 7770. Only when the Binary Loader is used to load other binary tapes is the SA:7777 used.

LOADING PROGRAMS

Introduction

When a computer is first manufactured, it "knows" nothing. It does not even know how to read in program tapes. The reading in of program tapes, called "loading," is accomplished using a fairly complex program called the Self-Checking Binary Loader. This program occupies locations 7632 - 7777₈, and once loaded should remain in memory permanently. All 1080 computers contain this program when shipped from the factory. The only conditions under which the Binary Loader must be reloaded are (a) if an experimental program runs wild or (b) if a power failure occurs while the 1080 is running.

Loading Programs Using the Binary Loader

Since the Binary Loader is self-checking, one can always start the computer at location 7777 and assume that if tape reads in, the loader is intact. If the computer halts when started at 7777, this indicates that the loader has been destroyed and it must be reloaded using Nico-Loadeon, as described in the next section.

To load a program tape using the Binary Loader:

- (a) Depress Wired Program STOP and Stored Program STOP to make sure the computer is not running.
- (b) Place the program tape, printed side up, in the tape reader. If you have a high speed reader, place the tape in the right-hand side and feed it through to the left-hand side. If you have only a low speed reader, set the reader switch to Free, place the tape in the reader, and turn the switch to Start.
- (c) Be sure that the power to the reader is turned on. For the high speed reader, this is an on-off switch on the front. For the low speed reader, turn the Teletype power switch to the Line position.
- (d) Set the switch register to 7777₈ (00 000 000 111 111 111 111). In this position, the right-hand twelve switches are up and the left eight switches are down.
- (e) Depress LOAD PC.
- (f) Press Execute
- (g) Depress CONTINUE
- (h) Press Execute.

The program should start reading in the binary tape. The Self-Checking Binary Loader automatically selects the correct tape reader. If the system contains a high speed reader, and the reader has tape in it, the program will be read from the high speed reader. If there is no high speed reader, or it contains no tape, the low speed reader will be used. If the program does not start, and the STOP light comes on, the Binary Loader has been destroyed and must be reloaded.

The Binary Loader program will halt under only two other conditions: (a) a checksum error, or (b) a rubout in the trailer of the tape. If the tape suddenly stops during read-in and the Teletype bell rings, a checksum error has been found. This indicates a tape reading error and means that the tape must be restarted at the beginning. Checksum errors are usually caused by torn or bent tape, tape loaded backwards, or occasionally, Teletype failure. Be sure to investigate the first two causes carefully before blaming the third. It is a good idea to duplicate all valuable tapes so that there is always a back-up copy available.

The only legal halt for the binary loader is upon finding a rubout (all 8 holes punched) in the trailer section of the tape. If the tape halts on a rubout while reading in the leader you have probably placed it in the reader backwards. Be sure to check the directional arrows printed on the tape before starting the Binary Loader. If the Binary Loader halts on a rubout, it may be restarted to read additional tapes by depressing Continue and pressing Execute.

Note that the Binary Loader is always started at 7777₈. The starting address printed on the tape label refers to the address at which the program is started once loaded. It does not refer to the Binary Loader.

Reloading the Binary Loader Using Nico-Loadeon

One could, of course, toggle in the entire Binary Loader at the switch register. However, this program is quite lengthy, occupying over 100 core locations, and this would be extremely tedious. A more efficient method is to write a shorter program, or "bootstrap" loader which then reads in the longer loading program. Nico-Loadeon utilizes this method twice. One first toggles in fourteen instructions and then reads in a two part tape through the low speed reader. The first part is read in using the toggled instructions and the second part using the program contained in the first section. When the second section is read in completely, the Self-Checking Binary Loader is resident and is used to read in all other tapes.

The fourteen instructions comprising Nico-Loadeon have been carefully designed to be entered with a minimum of switch register manipulation. Thus, in several cases, a number of switches stay the same between instructions, and in one case, an instruction is entered three times in succession.

The following instructions constitute the switch register portion of Nico-Loadeon. The Assembler mnemonic equivalents are given on the right, but are not needed to enter and use the program successfully.

	<u>Address</u>	<u>Contents</u>	<u>Assembler Equivalent</u>
	7736	7744	READ, R2
	7737	5007	LASH 7
S. A. =	7740	4453	RDTTY
	7741	6454	T1, TTYRF
	7742	1741	JMP T1
	7743	1001736	JMP @ READ
	7744	0171736	R2, ZERA
	7745	2705751	MMOM R4
	7746	2001736	JMS READ
	7747	2001736	JMS READ
	7750	2001736	R3, JMS READ
	7751	2407777	R4, ACCMZ 7777
	7752	1744	JMP R2
	7753	1750	JMP R3

To toggle in Nico-Loadeon, set the switch register to 7736 (00 000 000 111 111 011 110), depress LOAD PC and press Execute. The value 7736 will appear in the PC and the AC.

Then depress Deposit and Step, toggle in the instructions one by one, and press Execute to deposit each of them. Note that it is only necessary to load the first address into the PC, since Step automatically advances the location counter (PC) to the next address each time Execute is pressed. Thus, the contents of locations 7746-7750 can be entered by setting the switch register to 2001736 and pressing Execute three times in succession.

When you have toggled in all 14 instructions, go back and check to see that they have been entered correctly. This is accomplished by setting the switch register to 7736, depressing Load PC and pressing Execute. Then the locations are examined by depressing Examine while Step is depressed. The contents of a new memory location are displayed in the AC each time Execute is pressed. Since the Step button automatically increments the PC each time, the PC will always show an address one ahead of that being displayed.

When you are sure that the instructions have been entered correctly, place the Nico-Loadeon tape, printed side up, in the Teletype tape reader. The leader of this tape is entirely blank: it contains no punches along the right-hand side. Be sure that there is an inch or two of leader remaining before the first punched holes in the tape. Turn the reader to START and then start the computer at location 7740. This is accomplished by setting the switch register to 7740 (00 000 000 111 111 100 000), depressing Load PC, pressing Execute, depressing Continue and pressing Execute. (Be sure that you do not inadvertently press Start instead.)

The program should start and read in the tape. If the tape motion halts at any time, it indicates a program error. Go back, be sure that Nico-Loadeon is properly toggled in and start again.

Nico-Loadeon is self-modifying. This means that it will change as the tape reads in. When the tape has read in about one third of the way, the program will automatically change so that the section just read in is now in control and it reads in the rest of the tape.

When the tape has read in beyond all data holes, and the program is reading only trailer tape (containing holes along the right side only) the program may be stopped by turning off the tape reader and pressing Stop on the computer console. The Self-Checking Binary Loader is now loaded and can be started at 7777 to read in tapes, as described on page 1.

Binary Tape Format

Both the Intermediate and the Self-Checking Binary Loader utilize the same format of input tape. The only difference is that the longer loader uses the checksum information at the end of each section to check for read-in errors. The format is described below.

- (1) Leader - A row of column 7 (200_8) punches is used as leader and trailer. It must come before the first load information.
- (2) Data Format - Each 20-bit computer word is broken into three lines on paper tape, utilizing only columns 0-6. Column 7 is used to indicate a checksum and trailer. The word is broken up as follows:

Line 1	bits 19 - 14 (in tape columns 5-0)
Line 2	bits 13 - 7
Line 3	bits 6 - 0

The loader assembles each word from the three lines and adds it into a running sum, or "checksum."

- (3) Load Address - The first 20-bit word following the leader, or following each checksum, is the starting address for the data that follows. The load address is included in the checksum.
- (4) Data Words - Each 20-bit word following the load address is deposited in memory in sequential locations starting at the load address, and added into the checksum.
- (5) Checksum - At the end of each block of sequential data, the checksum is punched. It is the lowest order 20 bits of the running sum kept of that data block. It differs from actual load data only in that it has column 7 punched as well as columns 0-6. Following the checksum may be either a new load address or trailer code.

- (6) Trailer Code - This is identical to leader tape, except that it may have a Rubout punched in it. A rubout punched in pure trailer tape is a signal for the Binary Loader to halt.

Listing of the Loader Programs

Nico-Loadeon is listed on page 3. It operates by clearing the Teletype buffer and then skipping further reads at 7751 until a non-zero character is found. This is deposited at 7776. It then decrements the deposit address and deposits each assembled word in the next lower address. The last word is deposited in 7752, which causes a jump to the intermediate loader.

The intermediate loader, sometimes called the Short Binary Loader, is listed and explained below. The Self-Checking Binary Loader is listed following that.

/INTERMEDIATE BINARY LOADER
/CONTAINED IN NICO-LOADEON; NIC-80/S-7115B

*7752

```

7752      0  READ, 0  /TAPE READ SUBROUTINE
7753     5007    LASH 7 /SHIFT PREVIOUS AC CONTENTS
7754     6454  BACK, TTYRF /WAIT FOR TELETYPE FLAG
7755     1754    JMP BACK
7756     4453    4453    /"OR" TTY WITH AC
7757  1001752    JMP @ READ /EXIT READ SUBROUTINE

7760      0  NEXT, 0  /SUBROUTINE TO ASSEMBLE 20-BIT WORD
7761   170000    ZERA  /CLEAR AC
7762  2001752    JMS READ /READ ONE TTY CHARACTER
7763     2200    ANDZ (200 /IS THIS LEADER, TRAILER OR CHECKSUM?
7764     1770    JMP START /YES, GET NEW LOAD ADDRESS
7765  2001752    JMS READ /NO, GET REMAINING BYTES
7766  2001752    JMS READ
7767  1001760    JMP @ NEXT /AND EXIT

7770  2001760    START, JMS NEXT /GET NEW LOAD ADDRESS
7771  2405776    ACCM POINT /STORE IN POINTER LOCATION
7772  2001760    CONT, JMS NEXT /GET EACH DATA WORD
7773  3405776    ACCM @ POINT /AND STORE IT
7774  2125776    MPOM POINT /INCREMENT DATA WORD POINTER
7775     1772    JMP CONT /AND CONTINUE

7776      0  POINT, 0  /POINTER TO DATA LOCATION

```

/SELF CHECKING BINARY LOADER WITH AUTO-READER SELECTION

/SELF CHECKING BINARY LOADER WITH AUTO-READER SELECTION
/NIC-80/S-7115B-L

*7632

7632 0 READER, 0 /POINTER TO READER SUBROUTINE
7633 3052562 CM, 3052562 /SUM OF ADDRESSES 7634 - 7751

/INITIALIZE SELF CHECKER

7634 110115 START, MEMA (115 /INITIALIZE COUNTER
7635 2225721 ANGM CHEK /AND STORE NEGATIVE OF 115
7636 2165731 ZERM SUM /ZERO SUM
7637 2165734 ZERM HS /AND SUBROUTINES
7640 2165744 ZERM LS
7641 2111742 MEMA CA /SET POINTER TO START OF ADDRESSES
7642 2545733 AMOM POINT

/ADD TOGETHER CONTENTS OF LOCATIONS 7634-7751

7643 2135733 CL, MPOAM POINT
7644 3111733 MEMA @ POINT /GET CONTENTS OF EACH ADDRESS
7645 2515731 A+MAM SUM /AND ADD INTO SUM FOR SELF CHECKING
7646 2127721 MPOMZ CHEK /LAST ADDRESS?
7647 1643 JMP CL /NO, GET MORE
7650 2473633 A-MAZ CM /COMPARE WITH CORRECT SUM
7651 5220 HALT, STOP /HALT IF THEY DO NOT AGREE

/NOW SELECT THE PROPER READER

7652 166464 166464 /SKIP ON HSR FLAG & CLEAR AC
7653 110010 MEMA (10 /NO. OF ADDRESSES BETWEEN READ ROUTINES
7654 2511743 A+MA READ /MAKE HS INTO LS IF NO HSR
7655 2405632 ACCM READER /STORE CORRECT READ POINTER
7656 44453 RDTTY /CLEAR TTY FLAG AND START LOADING

7657 170000 LEAD, ZERA
7660 3001632 JMS @ READER /GET FIRST CHARACTER
7661 2200 ANDZ (200 /MUST HAVE LEADER TO LOAD
7662 162000 ZERZ
7663 1657 JMP LEAD /WAIT IN THIS LOOP UNTIL LEADER IS SENSED

7664 2175731 LOAD, ZERMA SUM /CLEAR CHECKSUM
7665 3001632 JMS @ READER
7666 462377 A-MZ (377 /RUBOUT?
7667 162000 ZERZ /SKIP
7670 1651 JMP HALT /YES, HALT
7671 2200 ANDZ (200 /LEADER?
7672 1664 JMP LOAD /YES, TRY AGAIN

7673 3001632 ADDRES, JMS @ READER
7674 3001632 JMS @ READER
7675 2405733 ACCM POINT /FIRST WORD IS LOAD ADDRESS
7676 2505731 NEXT, A+MM SUM /ADD INTO CHECKSUM
7677 170000 ZERA

/SELF CHECKING BINARY LOADER WITH AUTO-READER SELECTION

```

7700 3001632   JMS @ READER  /GET DATA WORDS
7701   2200   ANDZ (200   /LEADER-TRAILER-CHECKSUM?
7702   1710   JMP CHECK    /YES
7703 3001632   JMS @ READER  /NO, ASSEMBLE REMAINING BYTES
7704 3001632   JMS @ READER
7705 3405733   ACCM @ POINT  /AND STORE IN POINTER LOCATION
7706 2125733   MPOM POINT   /INCREMENT POINTER
7707   1676   JMP NEXT    /AND CONTINUE

7710 2001721   CHECK, JMS CHEK /READ IN CHECKSUM
7711 2001721   JMS CHEK     /MASKING OUT BIT 7
7712 2467731   A-MZM SUM   /DOES IT = CHECKSUM?
7713   1715   JMP ERROR    /NO, RING BELL AND HALT
7714   1664   JMP LOAD     /YES, LOOK FOR MORE SECTIONS

7715 110207   ERROR, MEMA (207 /BELL
7716   4443   PRTTY      /RING IT
7717 2111731   MEMA SUM   /PUT ERROR IN AC

7720   1651   JMP HALT    /AND HALT
7721   0      CHEK, 0    /SUBROUTINE TO READ IN CHECKSUM
7722   5210   CLL
7723   5150   EXCT AC0
7724   5204   STL      /SET LINK IF BIT 0=1
7725 3001632   JMS @ READER /READ IN ONE BYTE
7726   5101   SKIP L
7727 2011732   ANDA M200   /MASK OUT BIT 7 IF NON-DATA
7730 1001721   JMP @ CHEK   /AND EXIT

7731   0      SUM, 0    /CHECKSUM LOCATION
7732 3777577   M200, 3777577 /MASK FOR CHECKSUM WORDS
7733   0      POINT, 0   /DATA POINTER

7734   0      HS, 0    /HIGH SPEED READER SUBROUTINE
7735   5007   LASH 7    /SHIFT CURRENT AC
7736   6464   BACKK, HSRF /WAIT FOR FLAG
7737   1736   JMP BACKK
7740   4463   4463   /"OR" HSR WITH AC
7741 1001734   JMP @ HS    /AND EXIT
7742   7634   CA, START /POINTER FOR INITIALIZING SELF-CHECKER
7743   7734   READ, HS  /POINTER TO HIGH SPEED READER ROUTINE

7744   0      LS, 0    /LOW SPEED READER ROUTINE
7745   5007   LASH 7
7746   6454   L1, TTYRF /WAIT FOR TTY FLAG
7747   1746   JMP L1
7750   4453   4453   /"OR" TTY WITH AC
7751 1001744   JMP @ LS

*7777
7777   1634   JMP START  /POINTER TO START OF BINARY LOADER

```